# $Id: PKGBUILD 164989 2012-08-08 00:24:19Z tomegun $
# Maintainer : Ionut Biru <ibiru@archlinux.org>
# Contributor: Gabriel Martinez < reitaka at gmail dot com >

pkgname=libimobiledevice
pkgver=1.1.4
pkgrel=2
pkgdesc="Is a software library that talks the protocols to support iPhone and iPod Touch devices on Linux"
url="http://libimobiledevice.org/"
arch=('i686' 'x86_64')
license=('GPL2' 'LGPL2.1')
depends=('usbmuxd' 'gnutls' 'libgcrypt')
options=('!libtool')
source=(http://libimobiledevice.org/downloads/$pkgname-$pkgver.tar.bz2
	HOME-segfault.patch)

build() {
  cd "$pkgname-$pkgver"

  patch -p0 < "$srcdir/HOME-segfault.patch"

  PYTHON=/usr/bin/python2 ./configure --prefix=/usr
  make
}

package() {
  cd "$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
md5sums=('3f28cbc6a2e30d34685049c0abde5183'
         '0de709dbf8e79d0b00cca06f5e61d57c')
